

Storing multiple datasets in memory allows you to multitask, work with separate but related datasets, record results from one dataset into another, and more.

Work with separate but related datasets

You have two files, **persons.dta** and **counties.dta**, that are related. The persons live in the counties. You can load the datasets into separate frames and link them.

Open **persons.dta** in the **default** frame

```
. use persons
```

Create a new **counties** frame and open **counties.dta** in it

```
. frame create counties  
. frame counties: use counties
```

Link observations in the active frame (**default**) to the corresponding observations in the **counties** frame using variable **countyid**

```
. frlink m:1 countyid, frame(counties)
```

Copy variable **med_income** recording each county's median income from the **counties** frame to the active frame

```
. frget med_income, from(counties)
```

Create an alias for variable **med_homesize**, which records each county's median home size, so that you can use this variable as if it is in the active frame.

```
. fralias add med_homesize, from(counties)
```

Use frames to make your work easier

You have data for cities and countries around the world. You want to analyze the data for Germany efficiently without modifying your current data:

```
. frame put if country=="Germany", into(subset)  
. frame change subset  
. stata_commands  
. frame change default  
. frame drop subset
```

Record results in another frame

Create a new frame named **results** with variables **t** and **p**

```
. frame create results t p
```

Perform 1000 simulations, draw 100 random normal variates, perform a *t* test comparing the mean with 0, and post the *t* statistic and *p*-value into the **results** frame

```
. forvalues i=1(1)1000 {  
2.     quietly set obs 100  
3.     quietly generate x = rnormal()  
4.     quietly ttest x=0  
5.     frame post results (r(t)) (r(p))  
6.     drop _all  
7. }
```

Count observations in the **results** frame with *p*-values less than 0.05

```
. frame results: count p <= 0.05
```

Use commands or point and click

The screenshot displays the Stata software interface with several windows open. The main window shows a command history and a data table. The 'Frames Manager' window is open, showing a list of frames: 'persons', 'counties', 'default', 'health', 'insurance', 'persons', 'pop5', 'variables', 'states', and 'work'. The 'Rename existing frame' dialog box is open, showing 'default' as the frame to rename and 'census' as the new name. The 'Copy selected variables or observations to a new frame' dialog box is also open, showing 'pop5' as the name of the new frame and 'pop > 3000000' as the variables to copy. The main window shows a command history with the following commands: 1. usebase census, 2. frame rename default census, 3. describe, short, 4. frame put if pop > 3000000, into(pop5), 5. frame change pop5, 6. list state pop medage. The data table shows the following data:

state	pop	medage
1. California	23,667,992	29.90
2. Florida	9,766,324	26.70
3. Georgia	5,463,195	28.70
4. Illinois	11,426,558	29.90
5. Indiana	5,490,224	29.20
6. Massachusetts	5,232,837	31.20
7. Michigan	9,262,078	28.80
8. New Jersey	7,364,823	32.00
9. New York	17,558,072	31.90
10. N. Carolina	5,883,766	29.60
11. Ohio	10,797,630	29.60
12. Pennsylvania	11,803,895	32.10
13. Texas	14,229,191	28.20
14. Virginia	5,346,858	29.80